Solutions to the ATM  Database application:

Find the spot in the code where you should add your own code to answer these questions

1. What occurs in these lines of code: 
conCreditUnion = New OleDbConnection(ConnectString)

conCreditUnion.Open()

Ans: An instance of the OleDbConnection template is created and this object is stored as conCreditUnion. The Open method of conCreditUnion is run, which creates the connection to the database.

2. In this set of instructions, the .CommandText instruction contains a query that will act upon the database. What data do you expect to be returned to the data adapter?

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "SELECT * FROM tblCustomers"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

Ans: All records from tblCustomers

3. Add your own code to the Cancel button. Hint! you have to reset the Boolean flag that controls the PIN entry and have a suitable screen message

Ans:

blnPINEntry = True

        
lblScreen.Text = "Enter your PIN"

4. Display the Savings or Cheque balance to the screen after a transaction is made. Hint! Go to the Transaction subroutine to add your code there.

Ans:

             .Text = "Savings Balance: $" & sngSavingsBal

             .Refresh()

              WasteTime(4)

             .Text = "Cheque Balance: $" & sngChequeBal

             .Refresh()

              WasteTime(4)

5. Display the user's name after validation of the PIN acceptance. Hint! Write your code in the function blnValidPIN().
Ans:

'Display the user's name to validate the PIN acceptance

            With lblScreen

                .Text = rowCurrent.Item("Given") & " " & rowCurrent.Item("Family")

                .Refresh()

                WasteTime(4)

            End With
Solution Code:

Imports System.Data.OleDb

Public Class Form1

    '-----------------------------------------------------------------------------

    ' Copyright for this module belongs to the author. You may freely use, copy      

    ' and distribute for educational purposes only but not for commercial purposes.  

    '

    ' Author: Harry Schlanger

    ' Date: 2/07/07

    ' Location: Belmont High School

    ' Email: big_koala@hotmail.com

    ' Purpose: Introductory database material for Unit 4, Software Development

    ' Description: This program is an Automatic Teller Machine (ATM) system to 

    '              withdraw money from either Cheque strAccount or Savings strAccount. 

    '              Money has to exist in the strAccount. The ATM expects a four digit 

    '              PIN to be entered by the user. If accepted, it will ask for $$ 

    '              to withdraw. The balance will then be returned to the screen.

    '

    '              The interface is linked to an Access databse called CreditUnion.mdb 

    '              located in the \bin folder of the VB.Net project. If moved from 

    '              this location, a path to this database will need to be specified.

    '

    '              The software is based on the text by Graeme Summers 2003 Third Ed.

    '              "Programming with Visual Basic. Introduction to VB.Net"

    '              P176. Except that we have simplified it as much as possible AND 

    '              proper naming convention has been introduced. We have also placed

    '              internal comments to explain the workings of the code.

    '

    '-----------------------------------------------------------------------------

    'Declarations at module level

    Dim adpCustomer As OleDbDataAdapter

    Dim conCreditUnion As OleDbConnection

    Dim datCustomer As New DataSet()

    Dim strPIN, strAccount As String, sngChequeBal, sngSavingsBal As Single

    Dim blnPINEntry As Boolean = True, blnFirstDigit As Boolean = True

    Dim strWithDrawAmt As String = ""

    Private Function blnValidPIN() As Boolean

        '-------------------------------------------------------------------------------

        ' The key to opening the application is for this function to be true!!

        ' The dataset containing Customer table is searched for the PIN and the row

        ' of data containing that PIN is returned and saved in rowCurrent

        ' (It's equivalent to running a query but everything is done automatically

        '  behind the scenes by the ADO and OleDb technologies)

        ' If the row is found - the application is unlocked. If no row is found,

        ' the application is not opened.

        ' The value of the function is examined where the function is called. That is,

        ' in btnOK_Click()

        '---------------------------------------------------------------------------------

        Dim rowCurrent As DataRow = datCustomer.Tables("tblCustomers").Rows.Find(strPIN)

        Dim blnFound As Boolean = Not (rowCurrent Is Nothing)

        ' if data exists in the table then continue on...

        ' Get the field data from the record and store them in the program variables

        If blnFound Then

            sngChequeBal = rowCurrent.Item("ChequeBal")

            sngSavingsBal = rowCurrent.Item("SavingsBal")

            If rdbSavings.Checked Then

                strAccount = rowCurrent.Item("SavingsAcct")

            Else

                strAccount = rowCurrent.Item("ChequeAcct")

            End If

            'Display the user's name to validate the PIN acceptance

            With lblScreen

                .Text = rowCurrent.Item("Given") & " " & rowCurrent.Item("Family")

                .Refresh()

                WasteTime(4)

            End With

        End If

        Return blnFound

    End Function

    Private Sub WasteTime(ByVal N As Integer)

        ' The value of N is the number of seconds to waste

        Dim Start As Date

        Start = TimeOfDay

        Do

        Loop Until DateDiff(DateInterval.Second, Start, TimeOfDay) >= N

    End Sub

    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load

        '-------------------------------------------------------------------------------

        ' Connect to the database, then extract (or update) data from (to) the database

        ' via a data adapter named adpCustomer.

        ' The data is extracted (updated) using a SQL instruction located in the .CommandText 

        ' method of the data adapter's .SelectCommand method.

        ' The queried data is placed in the program's DataSet() named datCustomer

        ' for easy access in the program.

        ' Key(0) is an array with index 0 to store a column of data

        '------------------------------------------------------------------------------

        Dim Key(0) As DataColumn

        Dim ConnectString As String = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

        "Data Source=CreditUnion.mdb"

        conCreditUnion = New OleDbConnection(ConnectString)

        conCreditUnion.Open()

        adpCustomer = New OleDbDataAdapter()

        adpCustomer.SelectCommand = New OleDbCommand()

        ' use the data adapter to run a query against the database

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "SELECT * FROM tblCustomers"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

        ' place the data into the DataSet named datCustomer for easy access

        ' then return only data that is in the first column (with index 0)

        adpCustomer.Fill(datCustomer, "tblCustomers")

        Key(0) = datCustomer.Tables("tblCustomers").Columns(0)

        datCustomer.Tables("tblCustomers").PrimaryKey = Key

    End Sub

    Private Sub UpdateBalance()

        ' Here we have to interact with the database as we wish to update data using an

        ' UPDATE query. To test if this works, it's best to first create the query in Access,

        ' then get the SQL from Access and copy it here.

        '

        adpCustomer.SelectCommand = New OleDbCommand()

        With adpCustomer.SelectCommand

            .Connection = conCreditUnion

            .CommandText = "UPDATE tblCustomers SET SavingsBal = " + sngSavingsBal.ToString + ", ChequeBal = " + sngChequeBal.ToString + " WHERE PIN = '" + strPIN + "'"

            .CommandType = CommandType.Text

            .ExecuteNonQuery()

        End With

        ' Refresh the data - use the data adapter to place the new tblCustomers data into 

        ' the dataset datCustomer

        adpCustomer.Fill(datCustomer, "tblCustomers")

    End Sub

    Private Sub Transaction()

        Dim blnTooLow As Boolean

        blnTooLow = (rdbSavings.Checked And Val(strWithDrawAmt) > sngSavingsBal) Or (rdbCheque.Checked And Val(strWithDrawAmt) > sngChequeBal)

        With lblScreen

            If blnTooLow Then  ' not enough in strAccount

                .Text = "Balance too low"

                .Refresh()

                WasteTime(2)

            Else    ' enough to cover withdrawal

                .Text = "Transaction in progress"

                .Refresh()

                WasteTime(4)

                ' reduce amount in chosen strAccount

                If rdbSavings.Checked Then

                    sngSavingsBal -= Val(strWithDrawAmt)

                    .Text = "Savings Balance: $" & sngSavingsBal

                    .Refresh()

                    WasteTime(4)

                Else

                    sngChequeBal -= Val(strWithDrawAmt)

                    .Text = "Cheque Balance: $" & sngChequeBal

                    .Refresh()

                    WasteTime(4)

                End If

                UpdateBalance()

            End If

            blnPINEntry = True

            blnFirstDigit = True

            strWithDrawAmt = ""

            strPIN = ""

            .Text = "Enter your PIN"

        End With

    End Sub

    Private Sub btnOK_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnOK.Click

        Static intBadPINCount As Integer

        blnFirstDigit = True

        ' If the digits are entered and valid, a transaction occurs with the database

        ' However, if there is a problem with the PIN or if there are too many attempts

        ' at entering the PIN, then a message is returned to the user via the Screen.

        With lblScreen

            If blnPINEntry Then    ' entering PIN

                If blnValidPIN() Then

                    blnPINEntry = False

                    .Text = "Enter amount in $"

                Else    ' invalid PIN entry

                    strPIN = ""

                    intBadPINCount += 1

                    .Text = "Invalid PIN"

                    .Refresh()

                    WasteTime(2)

                    If intBadPINCount = 4 Then

                        intBadPINCount = 0

                        .Text = "Card taken, see manager"

                        .Refresh()

                        WasteTime(2)

                    End If

                    .Text = "Enter your PIN"

                End If

            Else    ' withdrawal amount entry (only multiples of $5)

                If Val(strWithDrawAmt) Mod 5 > 0 Then

                    .Text = "Invalid amount"

                    .Refresh()

                    WasteTime(2)

                    strWithDrawAmt = ""

                    .Text = "Enter amount in multiples of $5"

                Else    ' is a multiple of $5

                    ' modify the data in the database

                    Transaction()

                End If

            End If

        End With

    End Sub

    Private Sub btnClear_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnClear.Click

        blnFirstDigit = True

        If blnPINEntry Then    ' entering strPIN

            lblScreen.Text = "Enter your PIN"

            strPIN = ""

        Else    ' entering amount to withdraw

            lblScreen.Text = "Enter amount in $"

            strWithDrawAmt = ""

        End If

    End Sub

    Private Sub btnExit_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnExit.Click

        conCreditUnion.Close()

        End

    End Sub

    Private Sub btn1_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btn1.Click, btn0.Click, _

    btn2.Click, btn3.Click, btn4.Click, btn5.Click, _

    btn6.Click, btn7.Click, btn8.Click, btn9.Click

        ' At the start of the program both blnFirstDigit and blnPINEntry are set to TRUE

        ' The entered number is stored into strPIN using the system object sender.Text

        ' and an * is returned to the Screen. blnFirstDigit is set to FALSE so that next

        ' digit that is read does not wipe the Screen but instead shows ** and the digit

        ' accumulates into strPIN

        ' ...and so on

        If blnFirstDigit Then  ' starting entry

            lblScreen.Text = ""

        End If

        If blnPINEntry Then

            lblScreen.Text += "*"

            strPIN += sender.Text

        Else    ' entering amount to withdraw

            lblScreen.Text += sender.Text

            strWithDrawAmt += sender.Text

        End If

        blnFirstDigit = False

    End Sub

    Private Sub btnCancel_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles btnCancel.Click

        blnPINEntry = True

        lblScreen.Text = "Enter your PIN"

    End Sub

End Class

